NASA Technical Memorandum 100659

COMPUTER PROGRAMS FOR CALCULATION OF STING PITCH AND ROLL ANGLES REQUIRED TO OBTAIN ANGLES OF ATTACK AND SIDESLIP ON WIND TUNNEL MODELS

John B. Peterson, Jr.

July 1988

(NASA-TM-100659) COMPUTER PROGRAMS FOR CALCULATION OF STING PITCH AND ROLL ANGLES REQUIRED TO OBTAIN ANGLES OF ATTACK AND SIDESLIP ON WIND TUNNEL MODELS (NASA) 47 p N88-29820
CSCL 14B G3/09 Unclas 0166495

National Aeronautics and
Space Administration

Langley Research Center
Hampton, Virginia 23665-5225

INTRODUCTION

In many wind tunnels, the model support sting drive can both pitch and roll the sting and the model. This allows tests to be made in both angle of attack, α, and angle of sideslip, β, in the wind tunnel. Determination of the sting pitch and roll angles necessary to obtain desired α and β angles is fairly simple when the sting is straight. If the sting is offset or sting bending occurs, however, it can be very difficult to determine the sting pitch and roll angles to give the desired α and β on the model.

In order to solve this problem, a computer program has been developed to compute the pitch and roll position of a wind tunnel sting to position the model at the desired α and β with respect to the tunnel air stream for stings with offset angles in yaw, pitch and roll; and sting bending in yaw, pitch and roll. Computer programs that have been developed previously have been able to calculate the sting position for stings with one or two offset angles but generally they have not been able to accept sting angles in all three directions and/or stream flow angles. Also, previous programs usually generate the cosine of the sting pitch or roll angles and very elaborate methods are necessary to decide whether the pitch and roll are positive or negative angles in order to orientate the model correctly. The computer programs that are described in this report generate the sine or tangent of angles to reduce the ambiguity of whether the angles are positive or negative.

Two computer programs have been developed and are described in this report. These programs cover the case of no accelerometers on board the model and the case of accelerometers on board the model to measure model pitch and/or model roll with respect to gravity.

These programs are iterative programs in that they calculate the α and β of the model based on assumed pitch and roll angles of the sting and then calculate new positions for the sting pitch and roll to move the model closer to the desired α and β. This process is continued until the calculated α and β of the model match the desired α and β within a small increment. This procedure insures that the α and β calculated by the data reduction program based on the available information such as stream flow angles, sting bending due to measured model forces, and on board measurements of model pitch and roll, if available, will match the desired α and β.

Both of these programs accept three sting offset angles, three sting bending angles, and two tunnel flow angles. In addition, the second program accepts on board measured pitch angle and on board measured roll angle, if available.

SYMBOLS

u	Velocity along X axis of model
v	Velocity along Y axis of model
w	Velocity along Z axis of model
X	Longitudinal body axis, positive aft, see figure 1.
Y	Lateral body axis, positive to right, see figure 1.
Z	Vertical body axis, positive upward, see figure 1.
α	Model angle of attack, α = arctan (w/u), see figure 1.
β	Model angle of slideslip, β = arcsin ($-v/V_\infty$), see figure 1.
θ	Model pitch angle, positive direction is nose up, see figure 1.
θ_{off}	Sting offset angle in pitch
θ_{sb}	Sting bending in pitch
θ_{sc}	Sting pitch command
ϕ	Model roll angle, positive direction is right wing down, see figure 1.
ϕ_{off}	Sting offset angle in roll
ϕ_{sb}	Sting bending in roll
ϕ_{sc}	Sting roll command
ψ	Model yaw angle, positive direction is nose right, see figure 1.
ψ_{off}	Sting offset angle in yaw
ψ_{sb}	Sting bending in yaw

DETERMINATION OF ANGLE OF ATTACK AND ANGLE OF SIDESLIP FOR A WIND TUNNEL MODEL

The definition of angle of attach, α, on a wind tunnel model is given in reference 1 as the arctan (w/u) where w is the component of the free stream velocity along the Z axis of the model (vertical axis with the positive direction upward) and u is the component of the free stream velocity along the X axis of the model (longitudinal axis with the positive direction aft). This definition applies no matter what the orientation of the model is. By examining the signs of u and w, the correct quadrant for α can be determined between -180° and $+180^{\circ}$. If both u and w are zero, then the angle of attack is indeterminate, but in these programs, when both u and w are equal to zero, α is defined to be equal to zero.

The angle of sideslip, β, is defined in reference 1 as the arcsin $(-v/V\infty)$ where v is the component of the free stream velocity along the Y axis of the model (the lateral axis with the positive direction out the right or starboard wing) and $V\infty$ is the total free stream velocity. This means that β is positive when the flow is from the right.

In these computer programs, the free stream velocity, $V\infty$, is set equal to one, and the components of the free stream velocity along the wind tunnel axis system (u, v and w) are calculated from the angles of upwash and sidewash in the wind tunnel. The upwash angle (UWA) is defined to be positive when the flow is upward in the tunnel (i.e., the w component is positive) and the sidewash to be positive when the flow is from the right to the left (from the starboard to the port). For positive wind tunnel sidewash angle and the model pitch, roll and yaw angles equal to zero, the v component of the model velocity is negative and the model sideslip angle, β, is positive.

The three velocities (u, v and w) along the three axes of the model are recalculated for each rotation of the model. By convention, the rotations are taken in the order of yaw, pitch and then roll (i.e., rotation about the Z axis, the Y axis, and then the X axis of the model). The equations for the u, v and w velocities after rotation through each of the angles are given below.

Yaw (rotation about Z axis), ψ

$$UA = U * \cos(\psi) - V * \sin(\psi)$$

$$VA = V * \cos(\psi) + U * \sin(\psi)$$

$$WA = W$$

Pitch (rotation about Y axis), θ

$$UB = UA * \cos(\theta) - WA * \sin(\theta)$$

$$VB = VA$$

$$WB = WA * \cos(\theta) + UA * \sin(\theta)$$

Roll (rotation about X axis), ϕ

$$UC = UB$$

$$VC = VB * \cos(\phi) - WB * \sin(\phi)$$

$$WC = WB * \cos(\phi) + VB * \sin(\phi)$$

Where ψ is the angle of yaw (positive for nose right), θ is the angle of pitch (positive for nose up), and ϕ is the angle of roll (positive for right wing down). These angles are shown in figure 1 from reference 2.

The equations given above are used in subroutine ALPBET of program STNGOPR to calculate the velocities u, v and w after each rotation of the model. After the three velocities are determined, the angle of attack, α, and angle of sideslip, β, are calculated using the formulas given above.

DESCRIPTION OF THE COMPUTER PROGRAMS

Two computer programs were developed to calculate the sting pitch and roll angles necessary to position the model to obtain the desired α an β in the wind tunnel.

The first program, STNG, calculates the sting pitch and roll based on the sting geometry, such as sting offsets and sting bending. This program consists of the main program and two subroutines. A listing of the program is given in Appendix A.

The second program, STNGOPR, calculates the sting pitch and roll from the sting geometry but also uses inputs from accelerometers on board the model that measure the model pitch and roll angles. This program consists of a main program and four subroutines and a listing is given in Appendix B.

A list and description of the variables used in these programs is given after the description of the programs.

PROGRAM STNG

Program STNG calculates the sting pitch, θ_{sc}, and roll, ϕ_{sc} necessary to obtain a desired model angle of attack, α, and angle of sideslip, β. The program actually first calculates the α and β for an assumed sting pitch and roll. Then, by iteration, the assumed pitch and roll angles are changed until the calculated α and β agree with the desired or command α and β (ALPC and BETC).

Initial attempts to simple algorithms to determine which direction to move the sting to bring the model to the desired α and β, such as changing the sting pitch to change α and changing the sting roll to change β did not always converge to the desired α and β. Even attempts to control α by the sting movement (pitch or roll) that had the most influence on α and then controlling β by the remaining sting movement (roll or pitch) were not always successful, depending on sting geometry and the relation of the starting position of the sting to the final correct position of the sting.

To overcome these difficulties, two factors which are a combination of the errors in α ad β were developed for the program. These factors, called FT and FP are a linear combination of the errors in α and β (DALP and DBET) multiplied by a weighting factor which is proportional to the change in α and β due to a unit change in the sting pitch command (THESC) and the sting roll command (PHISC) respectively. The weighting factors for the function FT are FTA and FTB. FTA and FTB are proportional to the change in α and the change in β with a unit change in sting pitch. The function FT, then, is a linear combination of the weighting factor FTA times the error in α added to the weighting factor FTB times the error in β. Thus:

FT = FTA*DALP+FTB*DBET

After the function FT is found, the program calls subroutine CONV which uses two values of sting pitch position (THESC and THESCSV) and the two corresponding values of the function FT (FT and FTSV) to compute a new value for the sting position which should reduce the absolute value of FT. This subroutine uses a procedure, which is equivalent to Newton's method of finding the roots of an equation, to find the sting position that should reduce FT.

Each time subroutine CONV is called to find the sting position to minimize FT, subroutine VEL is called to determine the α and β at the new sting pitch angle. These new values for α and β are then used to calculate a new FT. This process is repeated until the absolute value of FT is less than .00001 (TOLF) or for a maximum of three iterations.

After FT is minimized, the weighting factor for the sting roll function (FP) are found. These weighting factors (FPA and FPB) are proportional to the change in α and the change in β, respectively, with a change in sting roll command (PHISC). The roll function, FP, then is:

FP = FPA*DALP+FPB*DBET

In the same manner as with FT, subroutine CONV is called to find a sting roll command (PHISC) that reduces the value of FP.

After FT and FP have been successively reduced, the entire procedure is repeated until the sum of the errors in α (ABS(DALP)) and β (ABS(DBET)) is less than the tolerance (TOLDAB) or for a maximum of six iterations.

In general, there are two positions of the sting drive that satisfy the command α and β. There are certain special cases where there are infinitely many positions and there are certain other cases where the sting cannot be positioned by pitch and roll drives to satisfy the commanded α and β. In those cases where there are two positions, they are nearly at a 180° rotation around the X axis of the wind tunnel to each other such that in one position the model is upright and in the other the model is inverted. The solution developed by the computer program can be either the upright or the inverted solutions depending upon the initial position of the sting and the α and β desired. The program checks the sting roll position of the solution to determine if it is outside the limits of -85° to 100°, and if so the program is rerun with the initial conditions of pitch equal to the negative of first solution pitch, and roll equal to the first solution roll ±180°. The solution with the new initial values for sting pitch and sting roll is very quick and the sting roll is within the limits of -85° to +100°. If an inverted solution is desired, the limits on the sting roll position can be changed to +80° for the lower limit and +280° for the upper limit.

The following is a list and description of the variables used in program STNG (all angles are degrees):

ALP   Angle of attack of model, α.

ALPC   Command or desired angle of attack.

BET   Angle of sideslip model, β.

BETC Command or desired angle of sideslip.

DALP Difference between angle of attack of the model and desired angle of attack.

DBET Difference between angle of sideslip of the model and desired angle of sideslip.

FP Function of sting roll (ϕ_{sc}) which is to be minimized by rolling the sting.

FPA Weighting factor for DALP in the function FP.

FPB Weighting factor for DBET in the function FP.

FT Function of sting pitch (θ_{sc}) which is to be minimized by pitching the sting.

FTA Weighting factor for DALP in the function FT.

FTB Weighting factor for DBET in the function FT.

PHIOFF Sting offset angle in roll (ϕ_{off})

PHISB Sting bending in roll (ϕ_{sb})

PHISC Sting roll command (ϕ_{sc}) required to position the model at the desired α and β.

PHISLL Lower limit on sting roll.

PHISLU Upper limit on sting roll.

PSIOFF Sting offset angle in yaw (ψ_{off}).

PSISB Sting bending in yaw (ψ_{sb}).

SWA Wind tunnel free stream sidewash angle, positive for flow from right.

THEOFF Sting offset angle in pith (θ_{off}).

THESB Sting bending angle in pitch (θ_{sb}).

THESC Sting pitch command (θ_{sc}) required to position the model at the desired α and β.

TOLDAB Convergence tolerance for the sum of the absolute values of DALP and DBET.

TOLF Convergence tolerance for the value of FT and FP.

UWA Wind tunnel free stream upwash angle, positive for flow from below.

u Free stream velocity component in the longitudinal direction in the wind tunnel (the total free stream velocity is assumed to be 1.0).

v Free stream velocity component in the lateral direction (flow from the right when looking forward is positive).

w Free stream velocity component in the vertical direction (upward flow is positive).

SUBROUTINE VEL

The purpose of subroutine VEL is to calculate the angle of attack, α, and angle of sideslip, β, of the wind tunnel model. The subroutine requires inputs of the velocity components in the wind tunnel u, v and w; the sting drive angles; the sting offset angles; and the sting bending angles. The subroutine calculates the components of the free stream velocity along the three axes of the model after each rotation angle using the formulas given in the section "Determination of Angle of Attack and Angle of sideslip for a Wind Tunnel Model." After the last rotation, these velocities are used to calculate the angle of attack, α, and angle of sideslip, β, using the following formulas:

$$\alpha = \arctan (w/u)$$

$$\beta = \arcsin (-v/V_\infty)$$

V_∞ is set equal to one in the main program (STNG) and therefore:

$$\beta = \arcsin (-v)$$

The following is a list and description of the additional variables used in subroutine VEL:

UB,...,UI	Longitudinal velocity in the model axis system after each rotation.
VB,...,VI	Lateral velocity in the model axis system after each rotation.
WB,...,WI	Vertical velocity in the model axis system after each rotation.

SUBROUTINE CONV

The purpose of subroutine CONV is to minimize a function Y = F(X). The function is calculated in the calling program and the subroutine calculates a new value of the independent variable X that will make the value of the dependent variable Y, nearer to zero. The new value of X is determined by calculating where a straight line through the last two previous pairs of points; X and Y, and XSAVE and YSAVE; intersects the X axis.

This new value is returned to the calling program in the X parameter and the old X and Y are placed in the XSAVE and YSAVE parameters at the end of the subroutine. In order to improve the stability of this procedure, certain limits are placed on the distance the new X value can be from the old X and XSAVE values. Line 14 of the subroutine CONV limits the new X value to be no more than 3.5 times the distance between the old X and XSAVE values away from the average of these values. Also if the two previous values for X are the same, the new X is calculated in a special way in statement 3. If the two previous Y values are the same, the new X is calculated to be the average of the two previous X's in statement 4.

The following is a list and description of the variables used in subroutine CONV:

DX	Absolute distance between X1 and X2.
X	Latest value of the independent variable. Also returned as the next value to be tried for the independent variable.
XA	Average of X1 and X2.
XK	Slope of line between X1, Y1 and X2, Y2.
XSAVE	Previous value of independent variable. Also returned as previous value of X.
X1	Previous value of the independent variable (same as XSAVE).
X2	Latest value of the independent variable.
Y	Latest value of the dependent variable.
YSAVE	Previous value of the dependent variable. Also returned as the previous value of Y.

Y1 Previous value of the dependent variable (same of YSAVE).

Y2 Latest value of the dependent variable.

PROGRAM STNGOPR

Program STNGOPR calculates the sting pitch and roll angles required to obtain the desired angle of attack and angle of sideslip in the wind tunnel when the model has on board measurements of the model pitch and roll (or pitch only if the roll is not available). The model pitch and roll is assumed to be measured relative to gravity.

At first it would appear that these cases would simplify the calculation of the alpha and beta of the model in the wind tunnel since the effects of sting offsets and sting bending are already included in the on board measurements of model pitch and roll. This is true in the case where a straight sting is used and the model yaw is assumed to equal zero. Since most wind tunnel tests use straight stings, on board measurements of model pitch and roll is a very valuable method of determining model alpha and beta in the wind tunnel. For bent stings, however, model yaw cannot be assumed to equal zero and model pitch and roll alone is not sufficient to determine model angle of attack and angle of sideslip.

In order to make program STNGOPR more generally applicable, no assumptions were made about the sting offset angles or the model yaw angle in the wind tunnel. Therefore, the program is applicable not only to the case of straight sting, but also to the case where the sting is offset and/or where sting bending occurs.

Program STNGOPR is similar to program STNG in that they both calculate the model α and β at a given sting position and then, by the use of a factor, try to reduce the difference between the calculated α and β.

Program STNGOPR uses the same functions (FT and FP) as program STNG uses to reduce the error between the calculated α and β and the desired α and β. The major difference between the two programs is that STNGOPR uses a three-step process to calculate α and β instead of the one step that STNG uses (subroutine VEL). The three steps are: first, calculate the yaw, pitch and roll angles of the model from the sting support system angles and the sting geometry (subroutine SIMUST), second, correct the pitch and roll of the model by the difference between the measured model pitch and roll, and the calculated model pitch and roll (DTHEMOB and DPHIMOB) determined at the beginning of the program, and third, calculate the α and β of the model from the model yaw, pitch and roll angles (subroutine ALPBET).

When programs STNG and STNGOPR are used in actual wind tunnel situations, the sting drive angles calculated by these programs will change somewhat as the sting and model move to the commanded positions because the sting bending angles will change as the angles of attack and sideslip of the model change and because the differences between the calculated model pitch and roll and the on board measured model pitch and roll change as the model attitude changes. The final position, however, will be the correct position to obtain the desired α and β, since the final values for the sting bending and measured model pitch and roll will be the same as those used by the wind tunnel data reduction program.

The following is a list and description of the additional variables used in program STNGOPR (all angles are in degrees):

DPHIMOB	Difference between PHIMOB and PHIMT determined at the beginning of the program.
DTHEMOB	Difference between THEMOB and THEMT determined at the beginning of the program.
PHIMOB	Roll of model as measured by on board accelerometers.

PHIMOBT	Theoretical on board roll of model. PHIMOBT = PHIMT + DPHIMOB
PHIMT	Theoretical roll of model as determined by subroutine SIMUST from the sting drive angles, sting offset angles, and sting bending angles.
PHIS	Actual sting drive roll angle.
PSIMT	Theoretical yaw angle of the model as determined by subroutine SIMUST from the sting drive angles, sting offset angles, and sting bending angles.
THEMOB	Pitch of model as measured by on board accelerometers.
THEMOBT	Theoretical on board pitch of model. THEMOBT = THEMT + DTHEMOB

THEMT — Theoretical pitch of model as determined by subroutine SIMUST from the sting drive angles, sting offset angles and sting bending angles.

THES — Actual sting drive pitch angle.

SUBROUTINE SIMUST

Subroutine SIMUST simulates the sting-support-sting-model system mathematically to calculate the model yaw, pitch and roll angles from the sting drive, sting offset and sting bending angles. The method of calculating the model yaw, pitch and roll is discussed in references 2 and 3, and is explained below.

The pitch angle of the model is determined by calculating the X component, in the model axis system, of a unit vector in the Z direction of the tunnel axis system (XZ). The pitch angle is then the arcsin (-XZ). The pitch angle can range from -90° to $+90^{\circ}$.

The roll of the model is determined by calculating the Y and Z components, in the model axis system, of a unit vector in the Z direction of the tunnel axis system (YZ and ZZ). The roll of the model is then the arctan (-YZ/ZZ) where the quadrant of the roll angle is determined by the signs of YZ and ZZ individually. The roll angle can range from -180° to 180°. If both YZ and ZZ are zero (i.e., the pitch angle is $\pm 90^{o}$) then the roll of the model is determined by the arctan (-YX, ZX) and the yaw of the model is defined to be equal to zero (where YX ad ZX are the Y component and the Z component respectively of a unit vector in the X direction of the tunnel axis system).

The yaw of the model is determined by calculating the X component in the model axis system of a unit Y vector in the wind tunnel axis system, XY, and the X component in the model axis system of a unit X vector in the tunnel axis system, XX. The yaw of the model is then the arctan (-XY/XX) and can range from -180^{o} to 180^{o}.

The following is a list and description of the additional variables used in subroutine SIMUST:

PHIS	Sting drive roll angle
THES	Sting drive pitch angle
XX, YX, ZX	The X, Y and Z components in the model axis system of a unit vector in the X direction in the tunnel axis system.

XY, YY, ZY — The X, Y and Z components in the model axis system of a unit vector in the Y direction in the tunnel axis system.

XZ, YZ, ZZ — The X, Y and Z components in the model axis system of a unit vector in the Z direction in the tunnel axis system.

SUBROUTINE COMP

Subroutine COMP is used to calculate the components, in the model axis system, of a unit vector in the wind tunnel axis system so that the model yaw, pitch and roll angles can be determined. This subroutine is very similar to subroutine VEL in program STNG which was used to calculate the velocity components in the model body axis system. Although this subroutine can calculate the X, Y and Z components, in the model axis system, of an arbitrary vector in the wind tunnel axis system, it is only used in this program to calculate the components of a unit vector in the X, Y or Z direction in the wind tunnel axis system. This means that one of the components, X, Y or Z, is set equal to one and the other components are set equal to zero in the calling program argument list. The components in the model body axis system of the unit vectors are then used in subroutine SIMUST to calculate the model yaw, pitch and roll.

The following is a list and description of the additional variables used in subroutine COMP:

X, Y, Z	X, Y and Z components of a vector in the tunnel axis system.
XB,...,XI	X component in the model axis system of a vector in the tunnel axis system after each rotation.
YB,...,YI	Y component in the model axis system of a vector in the tunnel axis system after each rotation.
ZB,...,ZI	Z component in the model axis system of a vector in tunnel axis system after each rotation.

SUBROUTINE ALPBET

The purpose of subroutine ALPBET is to calculate the angle of attack, α, and angle of sideslip, β, of the wind tunnel model. The subroutine requires inputs of the velocity components in the wind tunnel, U, V and W, and the model Euler angles, yaw (ψ), pitch (θ), and roll (ϕ). The subroutine calculates the components of the free stream velocity along the three axes of the model after each rotation angle using the formulas given in the section "Determination of Angle of Attack and Angle of Sideslip for a Wind Tunnel Model." After rotation through the three Euler angles, the velocities are used to calculate the angle of attack, α, and angle of sideslip, β, using the following formulas:

$$\alpha = \arctan (w/u)$$

$$\beta = \arcsin (-v/V\infty)$$

$V\infty$ is set to one in the main program (STNGOPR) and therefore:

$$\beta = \arcsin (-v)$$

The following is a list and description of the additional variables used in subroutine ALPBET (all angles are in degrees):

ALP	Angle of attack of model, α.
BET	Angle of sideslip of model, β.
PHIM	Angle of roll of model, ϕ.
PSIM	Angle of yaw of model, ψ.
THEM	Angle of pitch of model, θ.
UA, UB, UC	Longitudinal velocity component in the model axis system after each Euler angle rotation.
VA, VB, VC	Lateral velocity component in the model axis system after each Euler angle rotaton.
WA, WB, WC	Vertical velocity component in the model axis system after each Euler angle rotation.

CONCLUDING REMARKS

Two programs have been developed to calculate the pitch and roll angles of a wind tunnel sting drive system that will position a model at the desired angle of attack and angle of sideslip in the wind tunnel. These programs account for the effects of sting offset angles, sting bending angles and wind tunnel stream flow angles. In addition, the second program incorporates inputs from on board accelerometers that measure model pitch and roll with respect to gravity.

These program solve for the desired sting pitch and roll with an iterative procedure using the forward equations that calculate the model α and β from the sting geometry and the sting pitch and roll. This procedure avoids the ambiguity that is found in many inverse solutions that solve for the sting pitch and roll from the model α and β. Also, more sting offset angles, sting bending angles, and stream flow angles can be taken into account by using the forward equations.

A copy of the source code of these two programs can be obtained from the Langley Computer Center with the following statements:

GET, STNG/UN = 690250N

or GET, STNGOPR/UN = 690250N

The run times for these programs vary depending upon the number of iterations required to converge to a solution. When compiled and run under Fortran 5 (Fortran 77) on the Control Data Corporation Cyber CY180-860 computer at Langley Research Center, the run times for STNG is from 0.005 seconds for one iteration to 0.100 seconds for 36 iterations (the maximum allowed in these programs). The run times for STNGOPR range from 0.027 seconds to 0.352 seconds for one to 36 iterations respectively.

REFERENCES

1. Letter Symbols for Aeronautical Sciences. ASA Y10.7-1954, The American Society of Mechanical Engineers, 154.

2. Foster, Jean M. and Adcock, Jerry B., Users Guide for the National Transonic Facility Data System, NASA TM-100511, December 1987.

3. Fox, Charles H., Jr., Real Time Reduction Capabilities at the Langley 7- by 10-Foot High Speed Tunnel, NASA TM-78801, January 1980.

APPENDIX A

COMPUTER LISTING OF PROGRAM STNG

This appendix contains a computer listing of the program STNG which calculates the wind tunnel sting pitch and roll angles required to obtain the angle of attack, α, and angle of sideslip, β, on a wind tunnel model. The program accepts stream flow angles in two directions and sting offsets and sting bending in three directions.

```
      PROGRAM STNG (INPUT,OUTPUT)

C THIS PROGRAM CALCULATES THE STING DRIVE PITCH AND ROLL ANGLES TO
C POSITION A WIND TUNNEL MODEL AT THE COMMANDED ANGLE OF ATTACK (ALPC)
C AND ANGLE OF SIDESLIP (BETC).  THE PROGRAM ACCOUNTS FOR STING OFFSET
C ANGLES, STING BENDING ANGLES AND STREAM FLOW ANGLES.
C
C CODED BY -- JOHN B. PETERSON, JR.   NASA/LARC/TAD/NTFOB   1988

C              STING OFFSETS
      PSIOFF=0.
      THEOFF=45.
      PHIOFF=20.

C          STREAM FLOW ANGLES
C            POSITIVE FOR FLOW FROM BELOW AND FROM RIGHT
      SWA=.0
      UWA=.1
C               FREE STREAM VELOCITIES (TOTAL VEL. = 1.0)
      U=SQRT(  1./ (1.+(TAND(SWA))**2+(TAND(UWA))**2 )  )
      V=-U*TAND(SWA)
      W= U*TAND(UWA)

C               INITIAL VALUES
      THESC=0.0
      PHISC=0.0

C               LIMITS
      PHISLL=-85.
      PHISLU=100.
C FOR INVERTED RUNS USE FOLLOWING LIMITS
C     PHISLL=80.
C     PHISLU=265.

      TOLF=.00001
      TOLDAB=.00001

C               INPUTS TO CONTROL PROGRAM

C               COMMAND ANGLES
      ALPC= 5.
      BETC= 10.

C               STING DEFLECTIONS
      PSISB=0.4
      THESB=0.4
      PHISB=0.2

C              END OF INPUTS TO CONTROL PROGRAM
```

```
   50 CONTINUE

         PRINT 99
         PRINT 98,THESC,PHISC,ALPC,BETC,PSIOFF,THEOFF,PHIOFF

C              CONVERGE ON ALPC AND BETC

      DO 100 ICONV=1,6
         PRINT 96
         PRINT 97
      THESCSV=THESC-1.
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESCSV,PHISC,ALPSV,BETSV)
      DALPSV=ALPSV-ALPC
      DBETSV=BETSV-BETC
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISC,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FTA=ALP -ALPSV
      FTB=BET -BETSV
      X=1./(ABS(FTA)+ABS(FTB))
      FTA=FTA*X
      FTB=FTB*X
      FTSV=FTA*DALPSV+FTB*DBETSV
      FT  =FTA*DALP  +FTB*DBET
      DO 200 ICTHE=1,3
      CALL CONV(THESC,FT,THESCSV,FTSV)
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISC,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FT=FTA*DALP+FTB*DBET
         PRINT 98,THESC,PHISC,DALP,DBET,FT
      IF(ABS(FT).LT.TOLF) GO TO 210
  200 CONTINUE
  210 CONTINUE
      IF(ABS(DALP)+ABS(DBET).LT.TOLDAB) GO TO 1000
      IF(THESC.EQ.0.) THESC=.000001
         PRINT 96
         PRINT 97
      PHISCSV=PHISC-1.
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISCSV,ALPSV,BETSV)
      DALPSV=ALPSV-ALPC
      DBETSV=BETSV-BETC
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISC,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
```

```
      FPA=ALP-ALPSV
      FPB=BET-BETSV
      DENOM=ABS(FPA)+ABS(FPB)
      IF(DENOM.EQ.0.) GO TO 310
      X=1./DENOM
      FPA=FPA*X
      FPB=FPB*X
      FPSV=FPA*DALPSV+FPB*DBETSV
      FP  =FPA*DALP  +FPB*DBET
      DO 300 ICPHI=1,3
      CALL CONV(PHISC,FP,PHISCSV,FPSV)
      CALL VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISC,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FP=FPA*DALP+FPB*DBET
         PRINT 98,THESC,PHISC,DALP,DBET,FP
      IF(ABS(FP).LT.TOLF) GO TO 310
  300 CONTINUE
  310 CONTINUE
      IF(ABS(DALP)+ABS(DBET).LT.TOLDAB) GO TO 1000
  100 CONTINUE
 1000 CONTINUE

C CHECK TO SEE IF CONVERGED OUTSIDE PHIS LIMIT
      IF(PHISC.GT.PHISLU) THEN
         PHISC=PHISC-180.
         THESC=-THESC
         GO TO 50
      END IF
      IF(PHISC.LT.PHISLL) THEN
         PHISC=PHISC+180.
         THESC=-THESC
         GO TO 50
      END IF
         PRINT 96
         PRINT 95
         PRINT 98, THESC, PHISC, ALP, BET
      STOP
99    FORMAT("      THESC      PHISC     ALPC        BETC        PSIOFF" ,
     *        "     THEOFF     PHIOFF")
98    FORMAT (     8F10.5)
97    FORMAT("      THESC      PHISC      DALP        DBET        F")
96    FORMAT( )
95    FORMAT("      THESC      PHISC      ALP         BET")
      END
```

```
      SUBROUTINE VEL (U,V,W,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *                THESC,PHISC,ALP,BET)
      DOR=57.2957795
C             STING PITCH (Y)
      UB=U*COSD(THESC)-W*SIND(THESC)
      VB=V
      WB=W*COSD(THESC)+U*SIND(THESC)
C             STING ROLL (X)
      UC=UB
      VC=VB*COSD(PHISC)-WB*SIND(PHISC)
      WC=WB*COSD(PHISC)+VB*SIND(PHISC)
C             OFFSET YAW (Z)
      UD=UC*COSD(PSIOFF)-VC*SIND(PSIOFF)
      VD=VC*COSD(PSIOFF)+UC*SIND(PSIOFF)
      WD=WC
C             OFFSET PITCH (Y)
      UE=UD*COSD(THEOFF)-WD*SIND(THEOFF)
      VE=VD
      WE=WD*COSD(THEOFF)+UD*SIND(THEOFF)
C             OFFSET ROLL (X)
      UF=UE
      VF=VE*COSD(PHIOFF)-WE*SIND(PHIOFF)
      WF=WE*COSD(PHIOFF)+VE*SIND(PHIOFF)
C             STING BENDING IN YAW (Z)
      UG=UF*COSD(PSISB)-VF*SIND(PSISB)
      VG=VF*COSD(PSISB)+UF*SIND(PSISB)
      WG=WF
C             STING BENDING IN PITCH (Y)
      UH=UG*COSD(THESB)-WG*SIND(THESB)
      VH=VG
      WH=WG*COSD(THESB)+UG*SIND(THESB)
C             STING BENDING IN ROLL (X)
      UI=UH
      VI=VH*COSD(PHISB)-WH*SIND(PHISB)
      WI=WH*COSD(PHISB)+VH*SIND(PHISB)
C             ALPHA AND BETA
      IF(WI.EQ.0..AND.UI.EQ.0.)UI=.0000001
      ALP=ATAN2(WI,UI)*DOR
      IF(VI.LT.-1.)VI=-1.
      IF(VI.GT.1.) VI=1.
      BET=ASIN(-VI)*DOR
      RETURN
      END
```

```
      SUBROUTINE CONV (X,Y,XSAVE,YSAVE)
      X1=XSAVE
      Y1=YSAVE
      X2=X
      Y2=Y
      IF(X2.EQ.X1) GO TO 3
      IF(Y2.EQ.Y1) GO TO 4
      XK=(Y2-Y1)/(X2-X1)
      X=X2-Y2/XK
      XA=(X1+X2)/2.
      DX=ABS(X2-X1)
      IF(ABS(X-XA).GT.3.5*DX) X=XA+3.5*SIGN( DX, (X-XA) )
      GO TO 100
3     X=X2-Y2
      GO TO 100
4     X=(X2+X1)/2.
100   XSAVE=X2
      YSAVE=Y2
      RETURN
      END
```

APPENDIX B

COMPUTER LISTING OF PROGRAM STNGOPR

This appendix contains a computer listing of program STNGOPR which calculates the wind tunnel sting pitch and roll angles required to obtain the angle of attack, α, and angle of sideslip, β, on a wind tunnel model with on board accelerometers to measure model pitch and/or roll angles. The program accepts the accelerometer measurements of model pitch and roll, wind tunnel stream flow angles in two directions, and sting offsets and sting bending in three directions.

```
      PROGRAM STNGOPR(INPUT,OUTPUT)

C THIS PROGRAM CALCULATES THE STING DRIVE PITCH AND ROLL ANGLES TO
C POSITION A WIND TUNNEL MODEL AT THE COMMANDED ANGLE OF ATTACK (ALPC)
C AND ANGLE OF SIDESLIP (BETC).  THE PROGRAM ACCEPTS INPUTS FROM ACCELEROMETERS
C ON BOARD THE MODEL TO MEASURE THE MODEL PITCH AND ROLL RELATIVE TO GRAVITY
C AND IT ACCOUNTS FOR STING OFFSET ANGLES, STING BENDING AND STREAM FLOW ANGLES.
C
C CODED BY -- JOHN B. PETERSON, JR.   NASA/LARC/TAD/NTFOB   1988

C             STING OFFSETS
      PSIOFF=0.
      THEOFF=45.
      PHIOFF=20.

C         STREAM FLOW ANGLES
C           POSITIVE FOR FLOW FROM BELOW AND FROM RIGHT
      SWA=.0
      UWA=.1
C              FREE STREAM VELOCITIES (TOTAL VEL. = 1.0)
      U=SQRT(  1./ (1.+(TAND(SWA))**2+(TAND(UWA))**2 )  )
      V=-U*TAND(SWA)
      W= U*TAND(UWA)

C              INITIAL VALUES
      THESC=0.0
      PHISC=0.0
      THEMOB=0.0
      PHIMOB=0.0

C              LIMITS
      PHISLL=-85.
      PHISLU=100.
C FOR INVERTED RUNS USE FOLLOWING LIMITS
C     PHISLL=80.
C     PHISLU=265.

      TOLF=.00001
      TOLDAB=.00001

   50 CONTINUE

C              INPUTS TO CONTROL PROGRAM

C              COMMAND ANGLES
      ALPC=5.
      BETC=10.

C              STING DEFLECTIONS
      PSISB=0.4
```

```
      THESB=0.4
      PHISB=0.2

C             STING POSITION
      THES=THESC
      PHIS=PHISC

C             MODEL PITCH AND ROLL
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THES ,PHIS,PSIMOB,THEMOB,PHIMOB)
      THEMOB=THEMOB+0.
      PHIMOB=PHIMOB+0.

C            END OF INPUTS TO CONTROL PROGRAM


         PRINT 99
         PRINT 98,THESC,PHISC,ALPC,BETC,PSIOFF,THEOFF,PHIOFF

C              START OF CONTROL PROGRAM

C        DETERMINE THEORETICAL PITCH AND ROLL OF MODEL
C        AND COMPARE WITH MEASURED PITCH AND ROLL TO GET ERROR

      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THES ,PHIS,PSIMT,THEMT,PHIMT)
      DTHEMOB=THEMOB-THEMT
      DPHIMOB=PHIMOB-PHIMT

************************************************************
C IF THEMOB OR PHIMOB IS NOT AVAILABLE, DTHEMOB OR DPHIMOB
C SHOULD BE SET TO ZERO
C     DTHEMOB=0.
C     DPHIMOB=0.
************************************************************

C                CONVERGE ON ALPC AND BETC
          PRINT 96
          PRINT 94
          PRINT 98,THEMOB,PHIMOB,DTHEMOB,DPHIMOB

      DO 100 ICONV=1,6
         PRINT 96
         PRINT 97
      THESCSV=THESC-1.
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESCSV,PHISC,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALPSV,BETSV)
```

```
      DALPSV=ALPSV-ALPC
      DBETSV=BETSV-BETC
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC   ,PHISC,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FTA=ALP -ALPSV
      FTB=BET -BETSV
      X=1./(ABS(FTA)+ABS(FTB))
      FTA=FTA*X
      FTB=FTB*X
      FTSV=FTA*DALPSV+FTB*DBETSV
      FT  =FTA*DALP  +FTB*DBET
      DO 200 ICTHE=1,3
      CALL CONV(THESC,FT,THESCSV,FTSV)
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC   ,PHISC,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FT=FTA*DALP+FTB*DBET
         PRINT 98,THESC,PHISC,DALP,DBET,FT
      IF(ABS(FT).LT.TOLF) GO TO 210
  200 CONTINUE
  210 CONTINUE
      IF(ABS(DALP)+ABS(DBET).LT.TOLDAB  ) GO TO 1000
      IF(THESC.EQ.0.) THESC=.000001
         PRINT 96
         PRINT 97
      PHISCSV=PHISC-1.
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC,PHISCSV,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALPSV,BETSV)
      DALPSV=ALPSV-ALPC
      DBETSV=BETSV-BETC
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC   ,PHISC,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FPA=ALP-ALPSV
```

```
      FPB=BET-BETSV
      DENOM=ABS(FPA)+ABS(FPB)
      IF(DENOM.EQ.0.) GO TO 310
      X=1./DENOM
      FPA=FPA*X
      FPB=FPB*X
      FPSV=FPA*DALPSV+FPB*DBETSV
      FP   =FPA*DALP   +FPB*DBET
      DO 300 ICPHI=1,3
      CALL CONV(PHISC,FP,PHISCSV,FPSV)
      CALL SIMUST(PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *            THESC   ,PHISC,PSIMT,THEMT,PHIMT)
      THEMOBT=THEMT+DTHEMOB
      PHIMOBT=PHIMT+DPHIMOB
      CALL ALPBET(U,V,W,PSIMT,THEMOBT,PHIMOBT,ALP,BET)
      DALP=ALP-ALPC
      DBET=BET-BETC
      FP=FPA*DALP+FPB*DBET
         PRINT 98,THESC,PHISC,DALP,DBET,FP
      IF(ABS(FP).LT.TOLF) GO TO 310
  300 CONTINUE
  310 CONTINUE
      IF(ABS(DALP)+ABS(DBET).LT.TOLDAB   ) GO TO 1000
  100 CONTINUE
 1000  CONTINUE
C CHECK TO SEE IF CONVERGED OUTSIDE PHISC LIMIT
      IF(PHISC.GT.PHISLU) THEN
         PHISC=PHISC-180.
         THESC=-THESC
         GO TO 50
      END IF
      IF(PHISC.LT.PHISLL) THEN
         PHISC=PHISC+180.
         THESC=-THESC
         GO TO 50
      END IF
         PRINT 96
         PRINT 95
         PRINT 98, THESC, PHISC, ALP, BET
      STOP
99    FORMAT("     THESC     PHISC    ALPC      BETC      PSIOFF" ,
     *      "     THEOFF    PHIOFF")
98    FORMAT (     8F10.5)
97    FORMAT("     THESC     PHISC     DALP      DBET       F")
96    FORMAT( )
95    FORMAT("     THESC     PHISC     ALP       BET")
94    FORMAT("    THEMOB    PHIMOB   DTHEMOB   DPHIMOB")
      END
```

```
      SUBROUTINE SIMUST (PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *                   THES,PHIS,PSIMT,THEMT,PHIMT)

C THIS SUBROUTINE SIMULATES THE STING.  IT CALCULATES THE THEORETICAL MODEL
C YAW, PITCH AND ROLL GIVEN INPUTS OF THE
C STING OFFSETS, STING BENDING, AND STING PITCH AND ROLL.
      DOR=57.2957795

      CALL COMP (1.,0.,0.,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *           THES,PHIS,XX,YX,ZX)
      CALL COMP (0.,1.,0.,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *           THES,PHIS,XY,YY,ZY)
      CALL COMP (0.,0.,1.,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *           THES,PHIS,XZ,YZ,ZZ)

C       CALCULATE PITCH OF MODEL
      IF(XZ.LT.-1.)XZ=-1.
      IF(XZ.GT. 1.)XZ= 1.
      THEMT=ASIN(-XZ)*DOR

C            ROLL AND YAW OF THE MODEL
      IF(YZ.EQ.0..AND.ZZ.EQ.0.) THEN
C             CASE WHERE PITCH OF MODEL IS +/-90.
         PHIMT=ATAN2(-YX,ZX)*DOR
         PSIMT=0.
       ELSE
         PHIMT=ATAN2(-YZ,ZZ)*DOR
         PSIMT=ATAN2(-XY,XX)*DOR
      END IF
      RETURN
      END
```

```
      SUBROUTINE COMP   (X,Y,Z,PSIOFF,THEOFF,PHIOFF,PSISB,THESB,PHISB,
     *                   THES,PHIS,XI,YI,ZI)
C             STING PITCH (Y)
      XB=X*COSD(THES)-Z*SIND(THES)
      YB=Y
      ZB=Z*COSD(THES)+X*SIND(THES)
C             STING ROLL (X)
      XC=XB
      YC=YB*COSD(PHIS)-ZB*SIND(PHIS)
      ZC=ZB*COSD(PHIS)+YB*SIND(PHIS)
C             OFFSET YAW (Z)
      XD=XC*COSD(PSIOFF)-YC*SIND(PSIOFF)
      YD=YC*COSD(PSIOFF)+XC*SIND(PSIOFF)
      ZD=ZC
C             OFFSET PITCH (Y)
      XE=XD*COSD(THEOFF)-ZD*SIND(THEOFF)
      YE=YD
      ZE=ZD*COSD(THEOFF)+XD*SIND(THEOFF)
C             OFFSET ROLL (X)
      XF=XE
      YF=YE*COSD(PHIOFF)-ZE*SIND(PHIOFF)
      ZF=ZE*COSD(PHIOFF)+YE*SIND(PHIOFF)
C             STING BENDING IN YAW (Z)
      XG=XF*COSD(PSISB)-YF*SIND(PSISB)
      YG=YF*COSD(PSISB)+XF*SIND(PSISB)
      ZG=ZF
C             STING BENDING IN PITCH (Y)
      XH=XG*COSD(THESB)-ZG*SIND(THESB)
      YH=YG
      ZH=ZG*COSD(THESB)+XG*SIND(THESB)
C             STING BENDING IN ROLL (X)
      XI=XH
      YI=YH*COSD(PHISB)-ZH*SIND(PHISB)
      ZI=ZH*COSD(PHISB)+YH*SIND(PHISB)
      RETURN
      END
```

```
      SUBROUTINE ALPBET(U,V,W,PSIM,THEM,PHIM,ALP,BET)
      DOR=57.2957795
C                    YAW (Z)
      UA=U *COSD(PSIM)-V *SIND(PSIM)
      VA=V *COSD(PSIM)+U *SIND(PSIM)
      WA=W
C                   PITCH (Y)
      UB=UA*COSD(THEM)-WA*SIND(THEM)
      VB=VA
      WB=WA*COSD(THEM)+UA*SIND(THEM)
C                   ROLL (X)
      UC=UB
      VC=VB*COSD(PHIM)-WB*SIND(PHIM)
      WC=WB*COSD(PHIM)+VB*SIND(PHIM)
C               ALPHA AND BETA
      IF(WC.NE.0..AND.UC.NE.0.) THEN
        ALP=ATAN2(WC,UC)*DOR
       ELSE
        ALP=0.
      END IF
      IF(VC.LT.-1.)VC=-1.
      IF(VC.GT. 1.)VC= 1.
      BET=ASIN(-VC)*DOR
      RETURN
      END
```

```
      SUBROUTINE CONV (X,Y,XSAVE,YSAVE)
      X1=XSAVE
      Y1=YSAVE
      X2=X
      Y2=Y
      IF(X2.EQ.X1) GO TO 3
      IF(Y2.EQ.Y1) GO TO 4
      XK=(Y2-Y1)/(X2-X1)
      X=X2-Y2/XK
      XA=(X1+X2)/2.
      DX=ABS(X2-X1)
      IF(ABS(X-XA).GT.3.5*DX) X=XA+3.5*SIGN( DX, (X-XA) )
      GO TO 100
3     X=X2-Y2
      GO TO 100
4     X=(X2+X1)/2.
100   XSAVE=X2
      YSAVE=Y2
      RETURN
      END
```

PSI = ψ = Euler yaw angle = < ABC
THETA = θ = Euler pitch angle = < CBD (Note $\theta \neq \alpha$ unless ϕ = 0°)
PHI = ϕ = Euler roll angle = < CDE



Figure 1. Definition of Euler angles and directions

Report Documentation Page

National Aeronautics and Space Administration

1. Report No. NASA TM-100659	2. Government Accession No.	3. Recipient's Catalog No.
4. Title and Subtitle Computer Programs for Calculation of Sting Pitch and Roll Angles Required to Obtain Angles of Attack and Sideslip on Wind Tunnel Models		5. Report Date July 1988
		6. Performing Organization Code
7. Author(s) John B. Peterson, Jr.		8. Performing Organization Report No.
		10. Work Unit No. 505-60-21-04
9. Performing Organization Name and Address NASA Langley Research Center Hampton, VA 23665-5225		11. Contract or Grant No.
		13. Type of Report and Period Covered Technical Memorandum
12. Sponsoring Agency Name and Address National Aeronautics and Space Administration Washington, DC 20546-0001		14. Sponsoring Agency Code

15. Supplementary Notes

16. Abstract

Two programs have been developed to calculate the pitch and roll angles of a wind-tunnel sting drive system that will position a model at the desired angle of attack and angle of sideslip in the wind tunnel. These programs account for the effects of sting offset angles, sting bending angles and wind-tunnel stream flow angles. In addition, the second program incorporates inputs from on board accelerometers that measure model pitch and roll with respect to gravity.

The programs are presented in the report and a description of the numerical operation of the programs with a definition of the variables used in the programs is given.

17. Key Words (Suggested by Author(s))		18. Distribution Statement
Wind-tunnel apparatus	Computer program	Unclassified - Unlimited
Angle of attack	Roll	
Sideslip	Attitude control	Subject Category 09
Pitch	Mounts	
Yaw	Supports, Struts	

19. Security Classif. (of this report)	20. Security Classif. (of this page)	21. No. of pages	22. Price
Unclassified	Unclassified	46	A03

NASA FORM 1626 OCT 86

For sale by the National Technical Information Service, Springfield, Virginia 22161